
Xbase++ meets SQL (Using the ADS Database Engine)

Introduction

Xbase++ 1.9 has opened up some exciting new possibilities for incorporating new methods of database access into your existing Xbase++ applications while still leveraging your existing code base and databases.  New capabilities of the ADSDBE (Advantage Database Engine) from Alaska Software, and the Advantage Database Server from iAnywhere present us with a power and flexibility that is simple and affordable.

Several years ago, Extended Systems (now known as iAnywhere) developed a SQL system that could be used to manipulate ISAM (Foxpro and Clipper DBF) databases using standard SQL techniques.  This powerful feature of Advantage Server was mainly responsible for opening up the Advantage product line to a new world of possibilities.  Until now, these capabilities have been leveraged mostly by Delphi programmers who were originally Clipper programmers.  They discovered that Advantage provided them a solution to utilizing their existing ISAM (indexed sequential access method) databases, however they could no longer maintain their existing code base and have since moved away from the Xbase world.  This, of course, was before Xbase++ arrived on the scene.  If Xbase++ were available 3-4 years earlier, then we would be looking at a very different programming landscape than what we see today.  Instead of a Delphi/Advantage partnership that launched thousands more Delphi applications, we would be in the middle of an Xbase++/Advantage revolution that would have launched thousands more Xbase++ applications.

We cannot change the past, but we can certainly look toward a brighter future. That future continues to be in the hands of the Xbase++ development team, and in my opinion, they are moving in the right direction.  Although Xbase++ is not a mainstream language and Advantage Server is not a mainstream SQL server product, combining the power and flexibility of these products equals or surpasses the power of mainstream languages and SQL servers.

ISAM + SQL

It is well known that, over the years, SQL databases have become the choice of developers for industrial strength applications.  This is mostly due to the fact that popular SQL databases are also Client/Server databases and the Client/Server model affords the advantages of security, performance and low risk of corruption.  This capability was provided to the Xbase world many years ago by Extended Systems in their Advantage Server product, first made available to Clipper and then to Xbase++ by way of the ADSDBE.  SQL was not a consideration in the early days of Advantage server because the language concept of SQL was mostly incompatible with ISAM databases and Clipper/Xbase++ programmers became quite skilled at managing their data via techniques available in the Xbase language.  SQL (structured query language) is also a very different approach to data access than ISAM because it deals in data sets rather than providing the capability of browsing an entire database.  Many Xbase developers found this to be a very limiting factor in the design of an application and were convinced that their customers would reject a redesign that took away the power and performance that made Xbase/Clipper applications desirable.  With the development of more powerful ISAM indexing techniques, scoping, and browse navigation methods, Xbase++ applications proved to be much better performers than SQL-based applications when browsing data.

On the other hand, SQL provides a much simpler programming interface for complex data operations such as queries and updates and is extremely fast at providing small datasets as opposed to the common "filtering" technique.  It works well across multiple tables, provides a robust set of options and functions, does not require opening of databases and indexes, and does not require record locking.  It is nearly impossible to corrupt data tables when using SQL.

As bridges started to be built between the Xbase world and the SQL world, Clipper/Xbase and Delphi programmers found that a hybrid world was a much better place in which to live.  The developers at iAnywhere found an opportunity to give their community of programmers the best of both worlds and so they developed their "Streamline SQL" technology.  At the same time, they determined it to be in their interest to also develop a new proprietary database named ADT, so it could better accomodate the features of a more robust database, such as referential integrity, data triggers, etc.  ADT is still modeled after the ISAM (DBF) database, with its indexes, separate database files, record-based navigation techniques, etc.  They imagined that the Xbase development community would migrate their DBFs to ADTs to the gain the advantages that it could offer.

The jury is still out on this, however in my opinion, ADT is not a viable option for Xbase++ programmers.  It appears that the iAnywhere developers did not make it as compatible with existing applications as is necessary for a successful and simple migration.  This opinion came from several months of working with ADT and then abandoning all our efforts and refocusing back on maintaining our existing DBF databases.  After making this decision, our development effort was back on track and we made great progress in a very short time.  In this session, it is important to understand that we will be using only DBF/CDX data tables. No changes to existing databases are required to use SQL and Ads.

A little History

My experiences with SQL are probably not as broad as some Xbase++ programmers, yet I believe that I have enough experience to present a different approach to SQL and ISAM hybridization than what is most common.

I have seen or helped developed Xbase++ / SQL hybrid applications using a variety of techniques.  Here are a few scenarios.

1. Master Broker.  This was an application that consisted of a Clipper program and a VB program.  The Clipper program accessed DBF tables that were local to each Ticket Broker who used the software and the VB program accessed a MSSQL Server database that was shared by all the brokers.  The application has now been converted to 100% Xbase++ code and is used by ticket brokers to buy and sell tickets to each other and to the general public.  It uses FOXCDX to access the ISAM databases and SQLexpress to access the SQL tables.

2. Club Management.  This is an application that was converted from Clipper to Xbase++.  The developer wanted to provide his customers the ability to do SQL queries against any of his existing databases without changing any code in the application.  This was accomplished using Advantage SQL without the ADSDBE.  Instead the AdsSQL*() functions in the ACE32.DLL were used to create a SQL Query screen that displayed the results of SQL SELECT statements.

3. Integra.  This is an application that was converted from a FOXPRO 2.6 application to Xbase++.  It is an accounting application used by the Texas Oil industry.  It continues to use the same FoxPro databases and 95% of the code was just converted from text-based to GUI, but a new reporting/query system was created using AdsSQL*() functions.

4. AWolf.  This is a Enterprise-Level Clipper application that is used by several auto transmission remanufacturers thoroughout the U.S.  The Clipper application performed so well that the customers have been using it for many years and have been unsuccessful at replacing it with a third-party, 32-bit solution. The largest customer has over 40 stores in 5 states accessing the application through the internet running on Citrix and the manufacturing facility has another 150+ users accessing the application via LAN.  The application uses DBF/CDX database and Advantage Server. Several developers have offered to replace the application with a modern solution that uses MSSQL or Oracle SQL databases however they have not convinced the large customer that they can do this cost-effectively. It has been estimated that it will cost millions of dollars and several years to make this migration.

I was involved in helping to write the original Clipper application, much which is data-driven, and was asked by the developer to help him convince his largest customer that we can produce an Xbase++ application at a fraction of the cost that he has been quoted by others and that the application will leverage his existing databases and code base while giving him all the features offered to him by others.

We agreed that the application needed more relational integrity, SQL query capability, Crystal Reports compatibility, improved security and permissions system, and all the other improvements that come with a 32-bit GUI application, including using the internet for email and web-services.  The conversion to Gui has been ongoing for several years and is basically completed, however we had not looked very seriously at the database issues until we met with this customer and realized his concerns about SQL and integration with other third-party systems and developers.  For example, he had hired an ASP.NET programmer to write some custom applications and he was having much difficulty accessing the data.

For the remainder of this seminar, this Awolf application will be used to demonstrate how we leveraged Advantage "Streamline SQL" and "Data-Dictionary" features along with the new capabilities of ADSDBE 1.9 to open up exciting new possibilities for this legacy application.
The Advantage Client Library

The ADSDBE communicates with Advantage Server thru function calls to Ads*() functions in ACE32.DLL (Advantage Client Engine).  A file named ADS.PRG is provided on the conference disk and contains all the wrappers necessary to call any function in ACE32.DLL.  Simply include ADS.PRG in your application project file if you intend to use Advantage functions.  In most cases, very few function calls are necessary when using ADSDBE if all you intend to do is add SQL SELECT capability to your application.  If however, you intend to use SQL for more functionality, then a few more function calls are necessary.  These calls are covered in this seminar.  Any function starting with Ads* in the seminar text or in the sample programs are included in ADS.PRG.

iAnywhere also provides a FREE "Local Server".  The ADSLOC32.DLL can be used in place of the Client/Server software and it supports all the features discussed in this seminar.  This means that you can add Data-Dictionary and SQL support to your application without needing to have your customers purchase Advantage Server.  The ACE32.DLL Client engine is designed to communicate with a remote server or a local server.  An ADS.INI file is used to tell the Xbase++ application to connect to the local server if a remote  server is not found.   The local server makes it possible to develop the new  features and test them on a workstation that has no access to a remote server.
What's new in ADSDBE ?

The most important new feature is that ADSDBE now accepts cursor handles with the DbUseArea() function.

What this means is that you can now browse or otherwise manipulate data without changing your code or coding technique.

Here is a simple example:
oSession - A pointer to the AdsSession created by DacSession()

nCursor - A pointer to the SQL Cursor created by AdsExecuteSQLDirect().

DbUseArea( nil, oSession, '<CURSOR>' + L2Bin(nCursor) + '</CURSOR' )

Browse()

In order to use this new feature, you must first connect to an Advantage Data-Dictionary.  This is accomplished in the same way that you would normally make an ADSDBE connection except that you pass the name of the dictionary file in the connect string rather than the drive letter.

DbeSetDefault( "ADSDBE" )

cDictServer := Curpath() + '\XWOLF.ADD'

cSession := "DBE=ADSDBE;SERVER=" + cDictServer

oSession := DacSession():new(cSession)

Creating a workarea in this manner combines the best of the ISAM and SQL worlds by allowing existing indexes to be used with a SQL dataset, thereby allowing OrdSetFocus() to instantly change the order of records in the dataset.

The Data Dictionary (Why Use It?)

A Data Dictionary is the key to opening up an Xbase++ application to other applications and for providing SQL in the simplest possible form to Xbase++ programmers.

1. It combines the Xbase++ Work Area concept with the SQL SELECT statement.

2. It encapsulates the data and indexes of an application to improve performance when using SQL operations.

3. It provides a means for other applications, written in any language, i.e. ASP.NET, to access the Xbase++ application data via SQL statements.

4. It opens the application for use with third-party reporting tools such as Crystal Reports.

5. It provides an interface, via the ADS ODBC driver, to Microsoft applications such as Excel.

An Advantage Data-Dictionary is simply a file that contains information about every database and index table in an application.  The data-dictionary can be created using the Advantage Architect (ARC32.EXE) or in Xbase++ code as follows:

cDictName := 'ABC.ADD'

cDictDesc := 'My Test Application'

nError := AdsDDCreate( cDictName, .F., cDictDesc), @nHandle )

IF nError == 0

  nError := AdsConnect60( cDictName, ;

                ADS_REMOTE_SERVER + ADS_LOCAL_SERVER, ;

                'AdsSys', ;

                '', ;

                ADS_DEFAULT, ;

                @nHandle )

  IF nError == 0

    aDirectory := Directory('*.DBF')

    FOR i := 1 TO Len(aDirectory)

      cTableName := Strtran(Upper(aDirectory[i,1]),'.DBF','')

        nError := AdsDDAddTable( nHandle, cTableName, cTableName+'.DBF', ;

                                 ADS_CDX, ADS_OEM, '', '' )

    NEXT

  ENDIF

ENDIF

The Data-Dictionary is key to getting the most from Advantage SQL and the ADSDBE because dbUseArea() currently will not work with DacSession() objects that are FREE connections.

The above code example demonstrates creation of a DD that references DBF/CDX databases.  A DD can also contain references to DBF/NTX databases however, there is a bit more code necessary to add the the NTX files to the dictionary and associate them with the correct DBF files.  For simplicity, the examples in this session will use DBF/CDX databases.

It is important to note that very little application code needs to be changed when converting an application to using the ADSDBE.  Basically, the ADSDBE must be loaded at the start of the application, and a DacSession() must be created to connect the ADSDBE to Advantage Server.  File and index opening routines do not need to be touched.  Multi-threaded applications also require a small modification at the start of each thread to insure that the work area inherits the DacSession() object, but that's all there is to it.

The Advantage Architect

The Advantage Architect is a valuable resource for maintaining a data-dictionary and also for testing SQL SELECT statements and other SQL operations.  There are sufficient Ads*() functions in ACE32.DLL to write the Advantage Architect entirely in Xbase++ code, and in fact, I started to do that just to better understand the Ads*() functions.  I quickly learned, however, that I was only reinventing the wheel and decided to pursue more productive goals as we only had a few months to complete this project.

The architect can be used for SQL operations, modifying the data-dictionary, modifying the structure of databases and indexes, browsing/editing data tables, setting permissions, etc.  It is important to use this resource when converting an Xbase++ application to run from a Data-Dictionary.

SQL and the USE command

SQLUse() is a function uses the AdsStatement() class to create a SQL cursor from a statement and then apply it to a work area.

Example:

FUNCTION Test

LOCAL oStatement, cStatement

TEXT INTO cStatement WRAP

SELECT

invoice.invnmbr,

invoice.balance,

customer.custnmbr,

customer.billname,

customer.billstrt,

customer.billcity,

FROM customer, invoice

WHERE invoice.custnmbr = customer.custnmbr

ENDTEXT

oStatement := SQLUse( cStatement, 'INVOICES' )

Browse()

oStatement:close()

RETURN nil

*-----------

FUNCTION SQLUse( cStatement, cAlias )

LOCAL oStatement

oStatement := AdsStatement():New(cStatement,AdsSession())

IF oStatement:LastError > 0

  RETURN .f.

ENDIF

oStatement:Execute(cAlias)

RETURN oStatement

* ---------

CLASS AdsStatement

EXPORTED:

VAR Handle, Statement, Alias, Session, Cursor, LastError

INLINE METHOD GetLastError()

RETURN(::LastError)

* -------------

INLINE METHOD Init(cStatement, oSession)

IF(ValType(oSession)!="O")

  MsgBox( 'Parameter Type error : oSession' + Chr(13) + ;

          '(passed to AdsStatement:Init())' )

  ::LastError := 3

  RETURN Self

ENDIF

IF(!oSession:IsDerivedFrom("DacSession"))

  MsgBox( 'Parameter passed is not a DacSession : oSession' + chr(13) + ;

          '(passed to AdsStatement:Init())' )

  ::LastError := 4

  RETURN Self

ENDIF

::Session := oSession

RETURN ::Open(cStatement)

* ------------

INLINE METHOD Close()

IF(::HANDLE==NIL)

  RETURN(.F.)

ENDIF

IF (Used(::Alias))

  (::Alias)->(DbCloseArea())

ENDIF

::LastError := AdsCloseSQLStatement( ::HANDLE)

::Statement := NIL

::HANDLE    := NIL

::Alias     := NIL

RETURN .t.

* -------------

INLINE METHOD Open(cStatement)

LOCAL nH, nError, nErrorLen, cErrorString

IF ValType(cStatement)!="C"

  MsgBox( 'Parameter Type Invalid : Statement' + Chr(13) + ;

               '(passed to AdsStatement:Open())' )

  ::LastError := 1

  RETURN self

ENDIF

IF(Upper(Left(cStatement,Len(KEYWORD_SELECT)))!=KEYWORD_SELECT)

  MsgBox( 'Unsupported SQL statement' + Chr(13) + ;

          '(passed to AdsStatement:Open())' )

  ::LastError := 2

  RETURN self

ENDIF

::Statement := cStatement

nH := 0x0

::LastError := AdsCreateSQLStatement( ::Session:getConnectionHandle(), @nH )

::HANDLE := nH

IF ::LastError > 0

  cErrorString := _AdsGetLastError()

  MsgBox(cErrorString)

ELSE

  ::LastError := AdsVerifySQL( nH, cStatement )

  IF ::LastError > 0

    cErrorString := _AdsGetLastError()

    MsgBox(cErrorString)

  ENDIF

ENDIF

RETURN self

* --------------

INLINE METHOD Execute( cAlias )

LOCAL rc := 0x0, nCursor := 0x0, cErrorString, nErrorLen, nError

::LastError := AdsExecuteSQLDirect( ::HANDLE , ::Statement , @nCursor )

IF ::LastError > 0

  cErrorString := _AdsGetLastError()

  MsgBox(cErrorString)

  RETURN ''

ENDIF

DbUseArea( ,::Session, "<CURSOR>"+L2Bin(nCursor)+"</CURSOR>",cAlias)

IF (Used())

  ::Alias := Alias()

  ::Cursor := L2Bin(nCursor)

ENDIF

RETURN (::Alias)

ENDCLASS

* --------------

FUNCTION _AdsGetLastError()

LOCAL cErrorString, nErrorLen, nError

cErrorString := Space(500)

nErrorLen := 500

nError := 0

AdsGetLastError(@nError,@cErrorString,@nErrorLen)

cErrorString := Strtran(Pad(cErrorString,nErrorLen),';',Chr(13))

RETURN cErrorString

The ExecuteSQL() and ApplySQLParams() functions

ExecuteSQL() is a function that is used to execute any SQL statement, however it should be used only for statements that do not return a dataset but instead perform data operations that make updates to data tables, such as updating structures, global replaces, etc.  ExecuteSQL() can be used to execute statements embedded in code or to execute *.SQL files.

ApplySQLParams() is a function that is used to pass parameters into the SQL statement.

FUNCTION ExecuteSQL( cStatement, cStatus )

LOCAL nIndexMode := ADS_CDX, nHandle, nStatementHandle, nVerify, ;

      nLockingMode := ADS_PROPRIETARY, nCursorHandle

oSession := AdsSession()

nHandle := oSession:getConnectionHandle()

nStatementHandle := 0

AdsCreateSQLStatement( nHandle, @nStatementHandle )

AdsStmtSetTableType( nStatementHandle, nIndexMode )

AdsStmtSetTableLockType( nStatementHandle, nLockingMode )

nVerify := AdsVerifySQL( nStatementHandle, cSelect )

IF nVerify > 0

  cStatus := _AdsGetLastError()

  RETURN .f.

ENDIF

nCursorHandle := 0

AdsExecuteSQLDirect( nStatementHandle, cSelect, @nCursorHandle )

IF nCursorHandle = 0

  cStatus := _AdsGetLastError()

  RETURN .f.

ENDIF

RETURN .t.

* -----------

FUNCTION ApplySQLParams( cSqlStatement, aParams )

LOCAL i, cValue, cDate, xValue, nError

FOR i := 1 TO Len(aParams)

   IF aParams[i] == NIL

      LOOP

   ENDIF

   xValue := aParams[i]

   IF Valtype(xValue) == 'N'

      cValue := Alltrim(Str(xValue))

   ELSEIF Valtype(xValue) == 'L'

      IF xValue

         cValue := 'TRUE'

      ELSE

         cValue := 'FALSE'

      ENDIF

   ELSEIF Valtype(xValue) == 'D'

      cDate := DtoS(xValue)

      cDate := Ltrim(Rtrim(SubStr(cDate,1,4) + '-' + ;

         SubStr(cDate,5,2) + '-' + ;

         SubStr(cDate,7,2)))

      cValue := "{ d '" + cDate + "'}"

   ELSEIF Valtype(xValue) == 'C'

      cValue := "'" + xValue + "'"

   ELSE

      cValue := Trim(DC_XtoC(xValue))

   ENDIF

   cSqlStatement := StrTran(cSqlStatement,'?',cValue,,1)

NEXT

RETURN cSqlStatement

Using ExecuteSQL() to Modify a Database structure

Here is an example of a SQL file that is used to add a new field named RECORD_ID to all databases.

/* Record_ID.SQL */

ALTER TABLE APERAKT ADD COLUMN RECORD_ID Char( 20 )   ;

ALTER TABLE ARDUMP ADD COLUMN RECORD_ID Char( 20 )    ;

ALTER TABLE AUTODATA ADD COLUMN RECORD_ID Char( 20 )  ;

ALTER TABLE AUTOS ADD COLUMN RECORD_ID Char( 20 )     ;

ALTER TABLE BINLABEL ADD COLUMN RECORD_ID Char( 20 )  ;

ALTER TABLE BLDPOOL ADD COLUMN RECORD_ID Char( 20 )   ;

ALTER TABLE CAN_JOB ADD COLUMN RECORD_ID Char( 20 )   ;

.... 80 more tables

And here is the code that will execute this SQL file and modify the structure of every database in the Data Dictionary.

cSql := MemoRead('Record_Id.SQL')

cStatus := ExecuteSQL( cSql )

? cStatus

Using ExecuteSQL() to change Account numbers

Imagine a scenario in which your customer asked you to write a routine that would change all customer account numbers to the customer phone number.  This was a requirement in the Awolf application and we determined that using SQL was the only was to accomplish this task in a reasonable time-frame.  The ripple effect of changing customer account numbers on more than 200,000 customers was enormous and could take hours of programming plus hours more to run and test the code.  This is because every relational database with a customer account number also needed to be changed to maintain relational integrity.  In the Awolf application, we determined that 38 databases containing millions of records would be affected by this change.  We accomplished this task in a few hours of programming and less than 15 minutes of run time.  Here is how we did it:

First, we added a new field to ACCTS.DBF to store the old account number.

TEXT INTO cStatement

ALTER TABLE ACCTS ADD COLUMN OLD_ACCT Char( 10 )

ENDTEXT

ExecuteSQL( cStatement )

Next, we created a file named ACCTNO-UPDATE.SQL that contains the following SQL code:

BEGIN TRANSACTION ;

UPDATE ACCTS    SET old_acct = '%1' WHERE acct_no = '%1'  ;

UPDATE ACCTS    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE ACCTCROS SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE AUTOS    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE BLDPOOL  SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE CJPARTS  SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE CORES    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE CUSTDUMP SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE DELRO1   SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE DELROP   SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE DEPOSITS SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE GPTAGSPD SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE LEADS    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE LOSTSALE SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE MAILIN   SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE MAILLST  SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE NEWRO    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE RMA      SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE RMAPARTS SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE ORDERS   SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE PREMO    SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE PT       SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE PTPARTS  SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE RECOMREP SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE ROPARTS  SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE SALESTAX SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE TLCLIENT SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE TRANSM   SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE WARRANTY SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE COREBANK SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE MO       SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE TRANYOYO SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE SHIPMAST SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE SHIPACCT SET cust_acct = '%1' WHERE cust_acct = '%2' ;

UPDATE CONTACTS SET acct_no = '%1' WHERE acct_no = '%2'   ;

UPDATE CLAIMS   SET acct_no = '%1' WHERE acct_no = '%2'   ;

COMMIT WORK

Finally, we created a routine that steps through every record in ACCTS.DBF and executes the above code as follows:

cSQL := MemoRead( 'ACCTNO-UPDATE.SQL' )

USE ACCTS

nExceptions := 0

nUpdated := 0

DO WHILE !ACCTS->(Eof())

  cOldAcctNo := ACCTS->acct_no

  cNewAcctNo := Strtran(ACCTS->phone_no,'-','')

  cSQLUpdate := Strtran( cSQL, '%1', cNewAcctno )

  cSQLUpdate := Strtran( cSQLUpdate, '%2', cOldAcctNo )

  ? 'Updating ' + cOldAcctNo + ' to ' + cNewAcctNo

  IF Empty(cNewAcctNo)

    TL_Exception( 'ACCTNO_UPDATE_PHONE', cOldAcctNo, cNewAcctNo,,, ;

                  'Phone Number is Empty' )

    nExceptions++

  ELSEIF cNewAcctNo == cOldAcctNo

    // do nothing

  ELSEIF Len(cNewAcctNo) # 10

    TL_Exception( 'ACCTNO_UPDATE_PHONE', cOldAcctNo, cNewAcctNo,,, ;

                  'Phone Number is not 10 digits')

    nExceptions++

  ELSEIF ACCTS->(dbSeek(Pad(cNewAcctNo,10)))

    TL_Exception( 'ACCTNO_UPDATE_PHONE', cOldAcctNo, cNewAcctNo,,, ;

                  'Account Number Already Exists' )

    nExceptions++

  ELSEIF !ExecuteSQL( cSQLUpdate, @cStatus )

    TL_Exception( 'ACCTNO_UPDATE_PHONE', cOldAcctNo, cNewAcctNo,,, ;

                  cStatus )

    nExceptions++

  ELSE

    nUpdated++

  ENDIF

  ExecuteSQL( cSQLUpdate )

  ACCTS->(dbSkip())

ENDDO

ACCTS->(dbCloseArea())

? 'Accounts Updated = ', nUpdated

? 'Exceptions = ', nExceptions, 'Look at the EXCEPTION table'

Using ExecuteSQL() to add a new record to a database

In the AWolf application we created a new database to log exceptions.  Appending a record using SQL requires no record locking.  Here we use ExecuteSQL() and ApplySQLParams():

FUNCTION TL_Exception( cType, cRef1, cRef2, cRef3, cRef4, cErrorInfo )

LOCAL cStatement, aParams, cStatus

DEFAULT cType := 'UNKNOWN', ;

      cRef1 := '', ;

      cRef2 := '', ;

      cRef3 := '', ;

      cRef4 := '', ;

      cErrorInfo := ''

cRef1 := Strtran(cRef1,"'",'')

cRef2 := Strtran(cRef2,"'",'')

cRef3 := Strtran(cRef3,"'",'')

cRef4 := Strtran(cRef4,"'",'')

cErrorInfo := Strtran(cErrorInfo,"'","''")

TEXT INTO cStatement WRAP

INSERT INTO EXCEPTION ( type, ref1, ref2, ref3, ref4, errorinfo, date, time )

   VALUES ( ?, ?, ?, ?, ?, ?, ?, ? ) ;

ENDTEXT

aParams := { cType, cRef1, cRef2, cRef3, cRef4, cErrorInfo, Date(), Time() }

cStatement := ApplySQLParams( cStatement, aParams )

ExecuteSQL( cStatement, @cStatus )

RETURN nil

Using SQL files instead of Code

*.SQL files contain SQL commands that can be executed with the ExecuteSQL() function as follows:

cSQL := MemoRead('UPDATE.SQL')

ExecuteSQl( cSQL )

Many times it is much more convenient, many times more powerful and also much faster to write SQL code than to try to accomplish the same task in Xbase++ code.  In the Awolf application we were required to provide a more "normalized" solution to the master parts file (PARTS) where Department OnHand quantities needed to be placed into a new child database (PARTDEPT).  This involved creating a new table named PARTDEPT, adding new index tags to PARTDEPT, transferring the data from the PARTS table to PARTDEPT, then modifying the structure of PARTS by deleting and adding fields, then deleting the unused index tags in PARTS.

Another advantage to using SQL code is that the code can be run inside the Advantage Architect.  The application does not need to be running to write and test the SQL routine.  Once the routine is working as required, it can then be called in the application with ExecuteSQL().

Here is the SQL file that will accomplish this task:

/* PARTDEPT-CREATE.SQL

This script creates a new child table named PARTDEPT and tranfers

values in the following fields from PARTS to PARTDEPT:

 onhand -> onhand

 minlim -> minlim

 maxlim -> maxlim

 cost -> cost   // renamed this field in PARTS.DBF to std_cost

 onhand_xxx -> onhand

 minlim_xxx -> minlim

 maxlim_xxx -> maxlim

 cost_xxx -> cost

 renamed bin to std_bin */

CREATE TABLE PARTDEPT (

      PART_NO Char( 20 ),

      VENDOR Char( 4 ),

      DEPT_NO Numeric( 3, 0 ),

      ONHAND Numeric( 6 ,0 ),

      MINLIM Numeric( 6 ,0 ),

      MAXLIM Numeric( 6 ,0 ),

      COST Numeric( 11 ,3 ),

      BIN Char( 10 ),

      LSTYR_USED Numeric( 7, 0 ),

      YTD_USED Numeric( 7, 0 )) ;

EXECUTE PROCEDURE sp_CreateIndex(

   'PARTDEPT',

   'PARTDEPT.cdx',

   'PART_NO',

   'PART_NO+VENDOR+Str(DEPT_NO,3)',

   '',

   2,

   512 );

EXECUTE PROCEDURE sp_CreateIndex(

   'PARTDEPT',

   'PARTDEPT.cdx',

   'DEPT',

   'Str(DEPT_NO,3)+PART_NO+VENDOR',

   '',

   2,

   512 );

EXECUTE PROCEDURE sp_CreateIndex(

   'PARTDEPT',

   'PARTDEPT.cdx',

   'BIN',

   'Str(DEPT_NO,3)+BIN+PART_NO+VENDOR',

   '',

   2,

   512 );

INSERT INTO PARTDEPT

( PART_NO,

  VENDOR,

  DEPT_NO,

  ONHAND,

  MINLIM,

  MAXLIM,

  COST,

  BIN )

SELECT

  PART_NO,

  VENDOR,

  0,

  ONHAND,

  MINLIMIT,

  MAXLIMIT,

  COST,

  BIN

FROM PARTS

WHERE ONHAND > 0 OR MINLIMIT > 0 OR MAXLIMIT > 0 OR BIN IS NOT NULL;

INSERT INTO PARTDEPT

( PART_NO,

  VENDOR,

  DEPT_NO,

  ONHAND,

  MINLIM,

  MAXLIM,

  COST )

SELECT

  PART_NO,

  VENDOR,

  1,

  ONHAND_001,

  MINLIM_001,

  MAXLIM_001,

  COST_001

FROM PARTS

WHERE ONHAND_001 > 0 OR MINLIM_001 > 0  OR MAXLIM_001 > 0 ;

/* Duplicate "INSERT" above for 002 thru 030 */

DROP INDEX parts.part_oh   ;

DROP INDEX parts.part_bin  ;

DROP INDEX parts.part_bin2 ;

DROP INDEX parts.part_bin3 ;

DROP INDEX parts.part_low  ;

DROP INDEX parts.part_low2 ;

DROP INDEX parts.part_mxl  ;

DROP INDEX parts.pct_low   ;

DROP INDEX parts.pct_low2  ;

DROP INDEX parts.part_sld  ;

ALTER TABLE PARTS

   ALTER COLUMN cost std_cost Numeric( 9, 3)

   ALTER COLUMN bin std_bin Char( 10 )

   ADD COLUMN record_id Char( 20 )

   DROP ONHAND

   DROP ONHAND_001

   /* Duplicate "DROP" above for ONHAND_002 thru ONHAND_030 */

   DROP MINLIMIT

   DROP MINLIM_001

   /* Duplicate "DROP" above for MINLIM_002 thru MINLIM_030 */

   DROP MAXLIMIT

   DROP MAXLIM_001

   /* Duplicate "DROP" above for MAXLIM_002 thru MAXLIM_030 */

   DROP COST_001

   /* Duplicate "DROP" above for COST_002 thru COST_030 */

   DROP BIN2

   DROP BIN3

   DROP LASTYR

   DROP SOLD_YR_TD ;

Viewing the Tables in a Data-Dictionary

Having a browse or a tree-view of the tables in a data-dictionary is very convenient to both the application user and the programmer.  If you do not want to give your users access to the data-dictionary via Advantage Architect, then here is a routine that will display all the tables in a tree-view and allow the user to double-click the item to browse the table.  Simply substitute your own GUI browse for the Browse() function.  You may also add your own security to prevent editing or viewing without proper permissions.

FUNCTION TableTree()

LOCAL aTables, nDictHandle, GetList[0], GetOptions, oDlg, oTree, ;

      nEvent, mp1, mp2, oXbp

nDictHandle := AdsSession():getConnectionHandle()

aTables := _BuildObjectArray( ADS_DD_TABLE_OBJECT, nDictHandle, .t. )

ASort(aTables,,,{|a,b|a[1]<b[1]})

oDlg := XbpDialog():new(AppDeskTop(),,{0,0},{400,600}):create()

oTree := XbpTreeView():new(oDlg:drawingArea,,{0,0},oDlg:drawingArea:currentSize())

oTree:haslines := .t.

oTree:hasButtons := .t.

oTree:alwaysShowSelection := .t.

oTree:setFontCompoundName('9.Lucida Console')

oTree:itemSelected := {|a,b,o|_TableBrowse(o)}

oTree:create()

_BuildTree(oTree,aTables)

nEvent := 0

DO WHILE nEvent # xbeP_Close

  nEvent := AppEvent(@mp1,@mp2,@oXbp,1)

  IF nEvent > 0 .AND. Valtype(oXbp) == 'O'

    oXbp:handleEvent(nEvent,mp1,mp2)

  ENDIF

ENDDO

RETURN nil

* --------------

STATIC FUNCTION _TableBrowse( oTree )

LOCAL cTableName

cTableName := Substr(oTree:getData():caption,1,10)

USE (cTableName)

DC_CrtRun({||Browse()})

RETURN nil

* --------------

STATIC FUNCTION _BuildTree( oTree, aTables )

LOCAL i, cString, oItem

FOR i := 1 TO Len(aTables)

  cString := Pad(aTables[i,1],10) + ' : ' + Alltrim(aTables[i,2])

  oItem := oTree:rootItem:addItem( cString )

  oItem:cargo := aTables[i,1]

NEXT

RETURN nil

* --------------

STATIC FUNCTION _TableDesc( cTableName, nDictHandle )

LOCAL nProperty, xValue, nLen, nError

nProperty := ADS_DD_COMMENT

xValue := Space(200)

nLen := 200

nError := AdsDDGetTableProperty( nDictHandle, ;

                         cTableName, ;

                         nProperty, ;

                         @xValue, ;

                         @nLen )

RETURN Alltrim(xValue)

* --------------

STATIC FUNCTION _BuildObjectArray( nObject, nDictHandle )

LOCAL i, cData, nLen, nHandle, nError, aObject[0]

cData := Space(ADS_DD_MAX_OBJECT_NAME_LEN)

nLen := ADS_DD_MAX_OBJECT_NAME_LEN

nHandle := 0

nError := AdsDDFindFirstObject( nDictHandle, ;

                      nObject, ;

                      '', ;

                      @cData, ;

                      @nLen, ;

                      @nHandle )

IF nError == 0

  cData := Strtran(cData,Chr(0),'')

  AAdd( aObject, {Alltrim(cData),_TableDesc(Alltrim(cData),nDictHandle)})

ENDIF

DO WHILE nError == 0 .AND. nHandle > 0

  cData := Space(ADS_DD_MAX_OBJECT_NAME_LEN)

  nLen := ADS_DD_MAX_OBJECT_NAME_LEN

  nError := AdsDDFindNextObject( nDictHandle, ;

                      nHandle, ;

                      @cData, ;

                      @nLen )

  IF nError == 0 .AND. !Empty(cData)

    cData := Alltrim(Strtran(cData,Chr(0),''))

    AAdd( aObject, { Alltrim(cData), _TableDesc(Alltrim(cData),nDictHandle)})

  ENDIF

ENDDO

AdsDDFindClose(nHandle)

RETURN aObject

Accessing Application Data from Third-Party Apps

A significant advantage to using a Data-Dictionary in your application is the way it opens up your application data to other third party applications, such as ASP.NET, Crystal Reports, Visio, Excel, etc.

When the Advantage ODBC driver is installed, third-party applications that support ODBC will see your application data as an ODBC data source.

Conclusion

Xbase++ applications can be given new life and more power by utilizing inexpensive resources available in the ADSDBE from Alaska Software and Advantage Server from iAnywhere.

